

Streamline the Production and Archival Process for Study Milestones under Centralization Process

Yaoyu Zhu, Shuang Zhang, Sanofi

ABSTRACT

Building upon our previous work on the Centralization Process presented at PharmaSUG in 2023, this paper introduces an innovative solution to streamline the production and archival workflow for study milestones. The Centralization Process established a framework for managing multiple studies efficiently through centralized programs and documents. However, transferring all necessary elements (data, programs, macros, metadata, setup files, etc.) to the Production environment in the study milestone level remained a manual, time-consuming process prone to errors.

Our solution implements a one-click run process that automatically copies and archives all required elements from different folders to the Production environment, ensuring everything is transferred with the latest information from the centralization repository. This approach is particularly valuable when handling last-minute updates to ADaM datasets or TLFs requested by study statisticians or clinicians. By automating this critical step, we've dramatically improved the usability and acceptability of the centralization process, making preparation for production runs more efficient, reliable, and user-friendly.

The paper will demonstrate the implementation of this solution using SAS macros that manage backup, cleaning, and copying processes, providing practical examples of how this approach simplifies workflow while maintaining data integrity throughout the study milestone process.

INTRODUCTION

The centralization process is an integrated approach using centralized programs and documents to unify statistical analysis across multiple studies, especially in one project for a compound with similar study designs, sharing one central program and unique metadata while accounting for study specificities. The process involves determining scope of studies and programs to centralize, allocating responsibilities to programmers, managing centralized documents with dedicated selection methods for study differences, establishing a compound-level repository structure, executing programs through central macros, and managing study specificities through conditional programming.

The implementation of Centralization Process significantly improves consistency across studies, reduces effort when implementing changes, lowers error rates through multiple reviews, accelerates new study setup, facilitates data pooling activities, and enhances team collaboration and knowledge sharing.

However, the final step of preparing production environments remained a bottleneck, particularly when dealing with multiple studies sharing centralized resources, frequent updates to ADaM derivation rules, last-minute modifications to TLF requirements, and complex folder structures across different environments. The paper will introduce a solution to simplify the production and archival process for study milestones under the Centralization Process.

ENVIRONMENT AND WORKFLOW


DEV V.S. PROD

In this paper, we assume the system infrastructure consists of 2 parallel environments for each study analysis repository: one is **Develop Environment** (short as **DEV**), where the programs, metadata and documents are the dynamic working documents and can be updated in a real-time manner, the programs run with more flexibility in the development and test mode; the other one is **Production Environment** (short as **PROD**), where the final version of programs, metadata and documents should be archived and

the final official production run should be executed. Generally, when production run is completed and validated, all programs, documents and the generated datasets, outputs and other deliverables in PROD are locked and archived, which cannot be modified any more.

Except for some system set-ups, the folder structures are identical between **DEV** and **PROD** environments (see Figure 1). Under each environment of a study analysis folder, SDTM, ADS and REPORT folders include DATA, LOG, OUTPUT and PGM subfolders. The system browser provides a button to copy the corresponding programs and documents from DEV to PROD easily within the same study analysis folder.


Figure 1. Develop and Production Environments



CENTRALIZATION REPOSITORY STRUCTURE

Under Centralization Process (see Figure 2), a dedicated overall compound level repository should be created, where all the centralized programs, macros and documents are saved in the DEV Environment, e.g. the programs and the related macros to generate SDTM, ADaM and TLF. At each individual study level repository, only the batch run program will be created to call the centralized programs from overall compound level repository, as a result, the data and outputs will be produced at the study milestone level repository:

Figure 2. Repository Structure




CENTRALIZATION WORKFLOW

Generally, there're 3 repositories are involved in the centralization workflow (see Figure 3):

1. **DEV of Overall level repository:** all centralized programs, macros, metadata and documents will be generated, updated and saved. It's a centralized repository to maintain all programming elements for the involved centralized studies.
2. **DEV of Study Analysis level repository:** In development and test mode, a batch program will be created to call the centralized programs and macros to generate Study Analysis level data sets and outputs.
3. **PROD of Study Analysis level repository:** In production mode, the current version of the programs, macros, metadata to generate the deliverables for a dedicated milestone should be copied and saved in the PRODUCTION Environment of Study Analysis level repository. The production run should only use the programs and metadata from the same repository itself, which cannot be associated with the centralized overall repository anymore. This is to be compliant with the archival and traceability purpose for a formal statistical analysis.

Figure 3. Centralization Workflow



However, the system hasn't provided a function to copy the centralized elements (data, programs, macros, metadata, setup files, etc.) from the Development Environment of one repository to the Production Environment of another repository. We can only copy the programs one by one, folder by folder, which is a manual, duplicated and time-consuming process prone to errors for each study when it goes to the production run.

MACRO DEVELOPMENT

To integrate the manual clean-up and copy-paste processes into an automatic one-click solution prior to production run, we developed a set of SAS macros by wrapping the Linux command with the well-controlled execution location and set-ups for the target folder or target files.

First, a utility macro **%command** is created to execute system commands and capture their output.

```
%macro command(name,command);
    filename &name. pipe "
                        &command.
                        ";
    data _null_;
        infile &name. length = linelength;
        input line $varying512. linelength;
        put line=;
    run;
%mend;
```

Based on the utility macro **%command**, we initialize the function macros as below:

CLEAN-UP MACRO

m_clean_folder.sas

- Purpose: To remove all documents and subfolders from the selected folders.
- Before calling this macro, **PLEASE MAKE SURE** the targeted locations or folders are well archived to avoid any unplanned loss.
- The key steps are to check the input parameters, define the remove macro and utilize the macro to remove the selected folders:

```
/******
/*STEP 1 : INITIALIZATION
/******
/*Check and fill in the parameters*/
%if %length(&tgt_env.) eq 0 %then %do;
    options nosource;
    %put ERR%str(OR-)-----;
    %put ERR%str(OR-)ERROR: TGT_ENV is a required parameter, the value should be DEV or PROD;
    %put ERR%str(OR-)-----;
    options source;
    %goto EXIT;
%end;

%if %length(&tgt_study_name.) eq 0 %then %do;
    %let tgt_study_name=&w_study.;
%end;
%else %do;
    options nosource;
    %put ERR%str(OR-)-----;
    %put ERR%str(OR-)ERROR: It is not recommended to specify other TGT_STUDY_NAME than in this triplet;
    %put ERR%str(OR-)ERROR: Please make sure you run the the macro in the triplet to be cleaned ;
    %put ERR%str(OR-)-----;
    options source;
    %goto EXIT;
%end;
```

```

/*****
/*STEP 2 : Define the remove macro */
*****/
%macro rm_dir_file(dir,file);
%if %sysfunc(fileexist(%nrquote(&tgt_csa.&dir.))=1 %then %do;
  %command(remove,%str(rm -r &tgt_csa.&dir.&file.))
%end;
%mend;

/*****
/*STEP 3 : Remove the old backup folders with the same destination name to avoid mix */
*****/
%if %index(%upcase(&ADS.),%str(Y)) %then %do;
  %rm_dir_file(dir=%str(ADS/DATA/),file=%str(*));
  %rm_dir_file(dir=%str(ADS/LOG/),file=%str(*));
  %rm_dir_file(dir=%str(ADS/OUTPUT/),file=%str(*));
  %if %upcase(&ADS.) eq Y_ALL %then %do;
    %rm_dir_file(dir=%str(ADS/PGM/),file=%str(*));
  %end;
%end;

```

The macro parameters are as below:

```

%macro m_clean_folder(TGT_ENV           =
                        , TGT_STUDY_NAME =
                        , TGT_ANALYSIS_NAME=
                        , GLOBAL_CONTROL  =
                        , ADS              =
                        , DOC              =
                        , EXPLO            =
                        , MACRO            =
                        , METADATA         =
                        , QC               =
                        , REPORT           =
                        , SDTM             =
                        ) ;

```



```

/*-----*/
/* INPUT PARAMETERS : */
/* */
/* TGT_ENV          = Specify the target environment to perform clean folder*/
/*                  Required to be specified: DEV or PROD */
/* */
/* TGT_STUDY_NAME    = Specify the study name to perform clean folder */
/*                  Default value : &w_study */
/*                  Note: Other value is NOT permitted! */
/* */
/* TGT_ANALYSIS_NAME = Specify the analysis name to perform clean folder */
/*                  Default value : &w_analysis */
/*                  Note: Other value is NOT permitted! */
/* */
/* GLOBAL_CONTROL    = Flag to clean the whole triplet in pre-defined pattern*/
/*                  Null means no global pattern, other parameters below */
/*                  should be specified in details */
/*                  Y_ALL means to clean All folders and subfolders */
/*                  (except MISC) */
/*                  Y_PROD means to clean All folders and subfolders */
/*                  in PROD only (except MISC, METADATA, EXPLO) */
/*                  Y_SDTM means to clean DATA/LOG/OUTPUT in SDTM */
/*                  Y_ADS means to clean DATA/LOG/OUTPUT in ADS */
/*                  Y_REPORT means to clean DATA/LOG/OUTPUT in REPORT */
/*                  Default value : Null */
/* */
/* ADS              = Flag to clean ADS subfolders */
/*                  Null means not to be cleaned */
/*                  Y_EXCEPT_PGM means to clean DATA,LOG,OUTPUT */
/*                  Y_ALL means to clean DATA,LOG,OUTPUT,PGM */
/*                  Default value : Null */
/* */
/* DOC              = Flag to clean DOC subfolders */
/*                  Null means not to be cleaned */
/*                  Y_OTHER means to clean OTHER only */
/*                  Y_ALL means to clean all subfolders */
/*                  Default value : Null */
/* */
/* EXPLO            = Flag to clean EXPLO subfolders */
/*                  Null means not to be cleaned */
/*                  Y_EXCEPT_PGM means to clean DATA,LOG,OUTPUT */
/*                  Y_ALL means to clean DATA,LOG,OUTPUT,PGM */
/*                  Default value : Null */
/* */
/* MACRO            = Flag to clean MACRO folder */
/*                  Null means not to be cleaned */
/*                  Y_ALL means to clean all files in MACRO */
/*                  Default value : Null */
/* */
/* METADATA         = Flag to clean METADATA folder */
/*                  Null means not to be cleaned */
/*                  Y_ALL means to clean ADS,CT,SDTM subfolders */
/*                  Default value : Null */
/* */
/* REPORT           = Flag to clean REPORT subfolders */
/*                  Null means not to be cleaned */
/*                  Y_EXCEPT_PGM means to clean DATA,LOG,OUTPUT */
/*                  Y_ALL means to clean DATA,LOG,OUTPUT,PGM */
/*                  Default value : Null */
/* */
/* SDTM             = Flag to clean SDTM subfolders */
/*                  Null means not to be cleaned */
/*                  Y_EXCEPT_PGM means to clean DATA,LOG,OUTPUT */
/*                  Y_ALL means to clean DATA,LOG,OUTPUT,PGM */
/*                  Default value : Null */
/*-----*/

```

COPY MACRO

m_copy_folder.sas

- Purpose: To copy the selected documents or folders from the source repository to the target repository.
- Before calling this macro, **PLEASE MAKE SURE** the targeted locations or folders are well archived and cleaned to avoid any unplanned overwritten or loss.
- The key steps are to check the input parameters, define the copy macro and utilize the macro to copy the selected folders or files:

```
/******  
/*STEP 1 : INITIALIZATION */  
/******  
/*Check and fill in the parameters*/  
%if %length(&tgt_env.) eq 0 %then %do;  
    options nosource;  
    %put ERR%str(OR-)-----;  
    %put ERR%str(OR-)ERROR: TGT_ENV is a required parameter, the value should be DEV or PROD;  
    %put ERR%str(OR-)-----;  
    options source;  
    %goto EXIT;  
%end;  
  
%if %length(&tgt_study_name.) eq 0 %then %do;  
    %let tgt_study_name=&w_study.;  
%end;  
%else %do;  
    options nosource;  
    %put ERR%str(OR-)-----;  
    %put ERR%str(OR-)ERROR: It is not recommended to specify other TGT_STUDY_NAME than in this triplet;  
    %put ERR%str(OR-)ERROR: Please make sure you run the the macro in the triplet you want to copy to ;  
    %put ERR%str(OR-)-----;  
    options source;  
    %goto EXIT;  
%end;  
  
/******  
/*STEP 2 : Define the copy macro */  
/******  
%macro cp_dir_file(dir,file);  
%if %sysfunc(fileexist(%nrquote(&tgt_csa.&dir.)))=1 %then %do;  
    %command(copy,%str(cp -rp &src_csa.&dir.&file. &tgt_csa.&dir.))  
%end;  
%mend;  
  
/******  
/*STEP 3 : Copy the files */  
/******  
%if %index(%upcase(&ADS.),%str(Y)) %then %do;  
    %if %upcase(&ADS.) eq Y_DATA %then %do;  
        %cp_dir_file(dir=%str(ADS/DATA/),file=%str(*.sas7bdat));  
    %end;  
    %else %if %upcase(&ADS.) eq Y_PGM %then %do;  
        %cp_dir_file(dir=%str(ADS/PGM/),file=%str(*.sas));  
    %end;  
    %else %if %upcase(&ADS.) eq Y_ALL %then %do;  
        %cp_dir_file(dir=%str(ADS/DATA/),file=%str(*.sas7bdat));  
        %cp_dir_file(dir=%str(ADS/LOG/),file=%str(*.log));  
        %cp_dir_file(dir=%str(ADS/OUTPUT/),file=%str(*));  
        %cp_dir_file(dir=%str(ADS/PGM/),file=%str(*.sas));  
    %end;  
%end;
```

```

%if %length(&DES_FOLDER.) and %length(&DES_FILE.) %then %do;
  %if %sysfunc(fileexist(%nrbrace(&src_csa.&DES_FOLDER.&DES_FILE.)))=1
    and
    %sysfunc(fileexist(%nrbrace(&tgt_csa.&DES_FOLDER.)))=1
  %then %do;
    %cp_dir_file(dir=&DES_FOLDER.,file=&DES_FILE.);
  %end;
%end;

```

The macro parameters are as below:

```

%macro m_copy_folder(TGT_ENV           =
                    ,TGT_STUDY_NAME     =
                    ,TGT_ANALYSIS_NAME  =
                    ,SRC_ENV            =
                    ,SRC_STUDY_NAME     =
                    ,SRC_ANALYSIS_NAME  =
                    ,SRC_BACKUP_PATH    =
                    ,GLOBAL_CONTROL     =
                    ,ADS                =
                    ,DOC                =
                    ,EXPLO              =
                    ,MACRO              =
                    ,METADATA           =
                    ,QC                 =
                    ,REPORT             =
                    ,SDTM               =
                    ,DES_FOLDER         =
                    ,DES_FILE           =
                    );

```



```

/*-----*/
/* INPUT PARAMETERS : */
/*
/* TGT_ENV          = Specify the target environment to perform copy folder */
/*                  Required to be specified: DEV or PROD */
/*
/* TGT_STUDY_NAME    = Specify the target study name to perform copy folder */
/*                  Default value : &w_study */
/*                  Note: Other value is NOT permitted! */
/*
/* TGT_ANALYSIS_NAME = Specify the target analysis name to perform copy folder */
/*                  Default value : &w_analysis */
/*                  Note: Other value is NOT permitted! */
/*
/* SRC_ENV          = Specify the source environment to perform copy folder */
/*                  Default value : &w_MODWORK */
/*
/* SRC_STUDY_NAME    = Specify the source study name where the copy is from */
/*                  Default value : &w_study */
/*
/* SRC_ANALYSIS_NAME = Specify the source analysis name where the copy is */
/*                  from */
/*                  Default value : CSR */
/*
/* SRC_BACKUP_PATH   = Specify the source backup folder path where the copy */
/*                  is from */
/*                  Possible value: MISC/BACKUP_REGULAR/2022_04_06_Regular */
/*                  _DBE_17MAR2022 */
/*                  Default value : Null */
/*
/* GLOBAL_CONTROL    = Flag to copy the whole triplet in pre-defined pattern */
/*                  Null means no global pattern, other parameters below */
/*                  should be specified in details */
/*                  Y_ALL means to copy All folders and subfolders */
/*                  Y_PGM means to copy PGM only in SDTM, ADS, REPORT */
/*                  Y_DMC means to copy selected folders for DMC */
/*                  Default value : Null */
/*
/* ADS              = Flag to copy ADS subfolders */
/*                  Null means not to be copied */
/*                  Y_PGM means to copy PGM only */
/*                  Y_DATA means to copy only DATA */
/*                  Y_ALL means to copy DATA, LOG, OUTPUT, PGM */
/*                  Default value : Null */
/*
/* DOC              = Flag to copy DOC subfolders */
/*                  Null means not to be copied */
/*                  Y_OTHER means to copy OTHER only */
/*                  Y_METADATA means to copy _ref, _acronym, ADaM and CT */
/*                  metadata in DOC/OTHER */
/*                  Y_ALL means to copy all documents in all subfolders */
/*                  Default value : Null */
/*
/* EXPLO            = Flag to copy EXPLO subfolders */
/*                  Null means not to be copied */
/*                  Y_PGM means to copy only PGM */
/*                  Y_ALL means to copy DATA, LOG, OUTPUT, PGM */
/*                  Default value : Null */
/*
/* MACRO            = Flag to copy MACRO folder */
/*                  Null means not to be copied */
/*                  Y_ALL means to copy all files in MACRO */
/*                  Default value : Null */
/*

```

```

/* METADATA      = Flag to copy METADATA folder                                */
/*              Null means not to be copied                                    */
/*              Y_SETUP means to copy setup.sas program only                    */
/*              Y_ALL means to copy setup.sas and all files                      */
/*                  in ADS,CT,SDTM subfolders                                  */
/*              Default value : Null                                           */
/*              */
/* REPORT        = Flag to copy REPORT subfolders                             */
/*              Null means not to be copied                                    */
/*              Y_PGM means to copy only PGM                                    */
/*              Y_ALL means to copy DATA,LOG,OUTPUT,PGM                      */
/*              Default value : Null                                           */
/*              */
/* SDTM          = Flag to copy SDTM subfolders                               */
/*              Null means not to be copied                                    */
/*              Y_PGM means to copy only PGM                                    */
/*              Y_DATA means to copy only DATA                                */
/*              Y_ALL means to copy DATA,LOG,OUTPUT,PGM                      */
/*              Default value : Null                                           */
/*              */
/* DES_FOLDER    = [Optional] Specify a folder for the copy-paste              */
/*              Default value : Null                                           */
/*              */
/* DES_FILE      = [Optional] Specify the files for the copy-paste             */
/*              Default value : Null                                           */
/*              */
/*-----*/

```

PRE-PRODUCTION RUN

Based on the well-controlled clean-up and copy macros, we created a SAS program **pre_production.sas** to call the macros in adaptive patterns to meet the requirements of production run for a study milestone.

CLEAN UP PROD ENVIRONMENT

Call **%m_clean_folder** by filling in the parameters TGT_ENV with PROD and GLOBAL_CONTROL with Y_PROD to clean up all folders and subfolders in PROD.

```

%m_clean_folder(TGT_ENV      = %str(PROD),
                GLOBAL_CONTROL = %str(Y_PROD)
                );

```

TRANSFER FROM CENTRAL TO PROD

Call **%m_copy_folder** by multiple times filling with different parameters by indicating source repositories and selected folders and documents to copy all necessary elements from centralized overall repository or study analysis level DEV to study analysis level PROD.

*Copy all programs,macros,metadata from OVERALL/PHASE23 to study level PROD;

```
%m_copy_folder(TGT_ENV      = %str(PROD)
               ,SRC_ENV      = %str(DEV)
               ,SRC_STUDY_NAME = %str(OVERALL)
               ,SRC_ANALYSIS_NAME = %str(PHASE23)
               ,ADS           = %str(Y_PGM)
               ,DOC           = %str(Y_METADATA)
               ,MACRO         = %str(Y_ALL)
               ,REPORT        = %str(Y_PGM)
               ,SDTM          = %str()
               );
```

*Copy SDTM data from study level DEV to PROD;

```
%m_copy_folder(TGT_ENV      = %str(PROD)
               ,SRC_ENV      = %str(DEV)
               ,SRC_ANALYSIS_NAME = %str(DMC_2024_01)
               ,SDTM         = %str(Y_DATA)
               );
```

*Copy country_ref.sas7bdat to PROD;

```
%m_copy_folder(TGT_ENV      = %str(PROD),
               SRC_ENV      = %str(DEV),
               SRC_STUDY_NAME = %str(),
               SRC_ANALYSIS_NAME = %str(DMC_2024_01),
               DES_FOLDER     = %str(ADS/DATA/),
               DES_FILE       = %str(country_ref.sas7bdat)
               );
```

*Copy rnd_efcxxxxx.sas7bdat to PROD;

```
%m_copy_folder(TGT_ENV      = %str(PROD)
               ,SRC_ENV      = %str(DEV)
               ,SRC_STUDY_NAME = %str()
               ,SRC_ANALYSIS_NAME = %str(DMC_2024_01)
               ,DES_FOLDER     = %str(ADS/DATA/)
               ,DES_FILE       = %str(rnd_%lowercase(&w_study.).sas7bdat)
               );
```

*Copy pack programs to PROD;

```
%m_copy_folder(TGT_ENV      = %str(PROD)
               ,SRC_ENV      = %str(DEV)
               ,SRC_STUDY_NAME = %str()
               ,SRC_ANALYSIS_NAME = %str(DMC_2024_01)
               ,DES_FOLDER     = %str(REPORT/PGM/)
               ,DES_FILE       = %str(__pack_*.sas)
               );
```


*Copy logchecker_ads.sas and logchecker_report.sas to PROD;

```
%m_copy_folder(TGT_ENV      = %str(PROD)
               ,SRC_ENV      = %str(DEV)
               ,SRC_STUDY_NAME = %str()
               ,SRC_ANALYSIS_NAME = %str(DMC_2024_01)
               ,DES_FOLDER     = %str(QC/PGM/)
               ,DES_FILE       = %str(logchecker_ads.sas)
               );
%m_copy_folder(TGT_ENV      = %str(PROD)
               ,SRC_ENV      = %str(DEV)
               ,SRC_STUDY_NAME = %str()
               ,SRC_ANALYSIS_NAME = %str(DMC_2024_01)
               ,DES_FOLDER     = %str(QC/PGM/)
               ,DES_FILE       = %str(logchecker_report.sas)
               );
```

ONE-CLICK RUN

When all necessary elements are set up completely, we can easily implement a one-click run process that automatically copies and archives all required elements from different folders to the Study Analysis level Production environment, ensuring everything is transferred with the latest information from the centralization repository (see Figure 4).

Figure 4. Pre-Production Run Copy Map



This approach is particularly valuable when handling last-minute updates to ADaM datasets or TLFs requested by study statisticians or clinicians. E.g. After all statistical analysis including ADaM data sets and Tables, Listings Figures are generated in the Production Environment, that means all the programs, macros and metadata were archived and run in the PROD, an ad-hoc analysis is requested by the study team or some analysis should be adjusted after further discussion by the management team, which would lead to multiple modifications across the programs from ADaM level to TLF outputs level. According to the centralization process guideline, any modification in the program and metadata should be firstly implemented in the centralization repository, then the updated programs and metadata will be transferred again to the Production Environment to overwrite the previous version of the programs and metadata. With this one-click run solution by calling the function macros, we do not need to recheck and pick up how many programs are updated during the ad-hoc analysis, this is a time-consuming work because some analysis would involve modifications in multiple programs and the corresponding metadata due to the data set dependencies. We just need to click the batch run button to execute the **pre_production.sas** again, all the necessary elements (SDTM data, ADaM and TLF programs, macros, metadata, setup files, etc.), especially the latest version of updated ones, will be transferred to the Study Analysis level Production environment again without the needs to check the completeness one by one.

By automating this critical step, the synchronization between the centralization level repository and the study analysis level production repository becomes very fast and easy-to-control, we've dramatically improved the usability and acceptability of the centralization process, making preparation for production runs more efficient, reliable, and user-friendly, leading to be well adapted and applied to more projects and therapeutic areas.

BENEFIT AND RISK

BENEFIT

Efficiency and Time Savings

- Reduces manual effort by automating the transfer of programs, macros, and metadata
- Eliminates time-consuming copy-paste operations that previously required handling files one by one
- Accelerates preparation for production runs, particularly valuable during tight study milestone deadlines

Accuracy and stability

- Minimizes human errors associated with manual file transfers and selections
- Ensures complete transfer of all necessary elements without accidental omissions
- Maintains consistency between centralized repository and production environment

Adaptability to Change

- Facilitates rapid implementation of last-minute updates requested by statisticians or clinicians
- Enables seamless handling of ad-hoc analyses without extensive manual intervention
- Adapts to dynamic clinical trial analysis needs after 1st Production run

Scalability

- Standardizes production preparation process across projects and therapeutic areas
- Removes barriers to adopting the centralization process

RISK

System Dependencies

- Requires consistent folder structures and naming conventions across environments
- May need updates if system infrastructure changes
- Due to different system beyond the company, the design should be adjusted accordingly

Validation Challenges

- Automated processes need thorough validation to ensure they function as intended
- Changes to macros require revalidation
- When applied to a new project, a project level validation for setups is highly recommended

File Management Risks

- Automated processes could potentially overwrite some critical files if the parameters are incorrectly specified
- Requires careful parameter validation to prevent unintended data loss
- Needs proper backups before clean-up operations

Training Requirements

- Users must understand parameter settings to avoid errors

- Requires clear documentation and training for new team members
- May face resistance from those used to manual processes

By implementing appropriate controls, documentation, and training, these risks can be effectively mitigated. Specifically, establishing standard operating procedures for the use of these macros ensures consistent application across teams. Creating comprehensive user guides with examples helps new users understand parameter settings and avoid common pitfalls. With these measures in place, the statistical programming teams can fully leverage the efficiency gains, error reduction, and improved compliance offered by this streamlined approach, ultimately enhancing the quality and speed of clinical trial analysis under the Centralization Process.

CONCLUSION

The automated production and archival process represents a significant advancement in study milestone management under the Centralization Process. By eliminating manual steps and standardizing procedures, we've enhanced both efficiency and reliability in the clinical trial programming workflow. This solution complements our existing Centralization Process, creating a more robust and user-friendly framework for managing multiple clinical studies.

ACKNOWLEDGMENTS

We are sincerely acknowledged Gawain Gao and Sufang Huang for providing constructive comments with helpful discussion on this topic. We would also like to thank all the colleagues who are developing or using the centralization process in Sanofi.

RECOMMENDED READING

Shuang Zhang, Yaoyu Zhu. 2023. **Using Centralization Process to Streamline Programming Work on Multiple Late Phase Oncology Studies**[C]. *Proceedings of the PharmaSUG China 2023 Conference*. Nanjing, China. Available at <https://pharmasug.com.cn/resources/2023/2023-08/Pharmasug-China-2023-AP119.pdf>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Yaoyu Zhu
Sanofi
39F, Tower 3, No.1199, The north section of Tianfu Avenue, High-tech Zone, Chengdu, Sichuan 610041
Yaoyu.Zhu@sanofi.com

Shuang Zhang
Sanofi
6/F, HP Building, No.112, Jianguo Road, Chaoyang District, Beijing, 100022
Shuang2.Zhang@sanofi.com

Any brand and product names are trademarks of their respective companies.